FACT003

Care and Feeding of the PIC16C74 and Its Peripherals

Author: *Robert Angelo*
Microchip Technology Inc.

The PIC16C74 is one of the latest mid-range microcontrollers from Microchip Technology Inc. In this article we will be addressing a few of the new features and peripherals of this new part. The main focus will be on the A/D (Analog-to-Digital) Converter, the SCI (Serial Communication Interface), and the PWM (Pulse Width Modulator). Our intention is to give you a small program that initializes these peripherals as well as exercises them. A schematic is provided. The PICDEM™ 2 board from Microchip will run this program. The second trimpot does not exist on the PICDEM 2 board, so the second A/D value may float around. The second trimpot is only used to show a method of changing A/D input pins. If you are using the PICDEM 2 board, then the LED and a current limiting resistor must be connected to the PWM output. When the program is run, the RS-232 terminal will display two A/D values. The brightness of an LED is adjusted using pulse width modulation. The duty cycle is determined by the trimpot setting.

Assumptions

Although dangerous, sometimes we need to make assumptions. For this discussion on the PIC16C74, let us agree that RA0 and RA1 will be connected through a series resistor to the wipers on two potentiometers, with the other ends connecting across V_{DD} and ground (see schematic). The oscillator clock will be 4 MHz. First, we'll read an A/D input, send its result out the serial port (to be displayed on a PC terminal program), and then switch to the next channel. We will adjust the PWM output pulse width to match the first potentiometer. Each time we are ready to begin a new sequence, we will first send a pair of sync bytes to signal the receiving processor. To simplify our discussion, we will forgo using interrupts and we will do this in a polled fashion. The Watchdog Timer is disabled for this program.

To ensure there are no surprises, it is a good idea to initialize every Special Function Register (SFR) and data register to some known value prior to use.

A/D Converter Mysteries

The A/D converter and its eight input channels will be our first topic. Setting up the A/D converter involves two special function registers:

- ADCON0
- ADCON1

In the program included with this article, is a code segment `initad` that sets up the A/D. ADCON0 is the work horse register for this peripheral. This register is used to select the conversion clock frequency and channel. This register is also where we signal the start of a conversion and detect the completion of a conversion. ADCON1 has only one purpose in life for this part, and that is A/D port configuration. When ADCON1 is used, it does not override the TRISA register controls. The TRISA register must be set up. Once these registers are set up, all the program has to do is select the desired pin and set the GO/DONE bit in ADCON0. The program then waits for the conversion complete bit, GO/DONE, to be cleared by hardware. Then the ADRES (A/D conversion result register) register is read. The value from the first pot's conversion is then used to adjust the PWM pulse width, thereby adjusting the LED brightness.

FIGURE 1: PWM PULSE WIDTH



Pulse Width Modulation (PWM)

The PORTC<1> pin is used as the PWM output. The registers that need to be set up for this PWM operation are:

- TRISC
- T2CON
- CCPR2L
- PR2
- CCP2CON.

The code `initpwm` is an example of what might be done to initialize the PWM module. TRISC was cleared earlier, thus setting PORTC as output. By writing a "4" to the T2CON register, we will set the prescaler equal to 0 and select TIMER2 operation. Writing a 0Fh to the CCP2CON register selects PWM mode and standard resolution. The 0Fh written to the CCPR2L register sets the high period to a low value initially. Setting the PR2 register to FFh allows the CCPR2L value (from the A/D converter result) to approach a 100% duty cycle. Now we can control the brightness of the LED attached to this pin by adjusting the pot on pin RA0 and writing the A/D result to the CCPR2L register, as already described earlier.

SCI

The Serial Communications Interface Module is our RS-232 communications channel. We will configure the SCI as an asynchronous full duplex serial port. This is done with the routine at `initsci` in the program provided. There are a few fine points to remember relative to this peripheral. The baud rate is determined by a dedicated eight-bit baud rate generator and can be used to derive standard baud rate frequencies from the oscillator. Since we are not using interrupts, there are only five registers to deal with:

- RCSTA - receiver status
- TXSTA - transmitter status
- TXREG - transmit buffer
- RCREG - receive buffer
- SPBRG - to set the baud rate generator

FIGURE 2: SERIAL COMMUNICATIONS INTERFACE MODULE



First global interrupts are disabled. The `initsci` code does the serial port setup and the `sendat` code handles the actual sending of the data.

The SCI is setup for 2400 baud, 8 data bits and 1 STOP bit with no parity. A terminal program, such as TERMINAL in Windows®, set to the same settings can be used to see our output. If you use the Windows TERMINAL program, then set the communications parameters to 2400 baud, 8 data bits, 1 STOP bit, no parity and hardware handshake.

Tying The Pieces Together

The main loop for getting the process running and restarting it again is `mloop`. The `adcnvrt` routine handles port pin selection and actual conversion control. The `dopwm` routine handles updating the PWM duty cycle register CCPR2L. The routine `sendat` checks transmit ready status and loads the transmit buffer when the status reports ready. You will notice there is no error recovery routine. It is up to the user to determine.

Here is what the program will do:

Once all peripherals have been initialized, two sync bytes "< >" are sent to the terminal. The A/D conversion results are then sent and the LED brightness is adjusted to match the RA0 trimpot setting. To simplify displaying A/D values, only the highest nibble is used, and thirty is added to put it into an ASCII range.

APPENDIX A: CARE AND FEEDING OF PIC16C74 SOURCE CODE

```
LIST P=16C74
INCLUDE P16CXX.INC              ;new include file that comes with MPASM (on BBS)
Adcnt   equ     20h             ;a/d converter pin count register
Adcntw  equ     21h             ;a/d converter pin work register
Temp    equ     22h             ;temporary data holding register seems we always need one
        org     0
        goto    init            ;go to where our code really begins
        org     5h              ;begin program above interrupt service vector address
init    bcf     INTCON,7        ;make sure we don't get interrupted
        clrf    PORTA           ;don't rely on anything, set port latches where you want them
        clrf    PORTB
        clrf    PORTC
        clrf    PORTD
        clrf    PORTE
        clrf    Adcnt           ;clear RAM registers we will be using
        clrf    Adcntw
        clrf    Temp
        bsf     STATUS          ,RP0 ;switch to page 1 to access trisX registers
        clrf    TRISB           ;set all ports outputs
        clrf    TRISC           ;just for this program to minimize current
        clrf    TRISD           ; and prevent pins from floating
        clrf    TRISE
        movlw   0Bh
        movwf   TRISA           ;set analog inputs as inputs, the rest as outputs
        bcf     STATUS,RP0      ;
initad  movlw   0C1h            ;Internal RC A/D clock, input channel 0 , A/D on
        movwf   ADCON0          ;(user must wait for specified period before sampling)
        bsf     STATUS,RP0      ;select page 1 of the SFRs
        movlw   4
        movwf   ADCON1          ;setup a/d inputs on RA0, RA1 and RA3 with Vref = Vdd
                                ;we are still in page 1 of the SFRs
initsci movlw   19h             ;setup 2400 baud
        movwf   SPBRG
        movlw   20h             ;setup for async operations
        movwf   TXSTA
        bcf     STATUS,RP0      ;back to page 0 for a moment
        movlw   80h             ;enable serial port operations and the associated pins
        movwf   RCSTA
        clrf    TXREG           ;clear our serial port buffers for start up
        clrf    RCREG
initpwm movlw   4h              ;setup T2CON with prescaler = 0 and timer2 on
        movwf   T2CON
        movlw   0fh             ;setup capture/compare to PWM mode standard resolution
        movwf   CCP2CON
        movlw   0fh             ;set compare register to half for now
        movwf   CCPR2L
        bsf     STATUS,RP0      ;select page 1 for the PR2 register
        movlw   0ffh
        movwf   PR2
```

```
        bcf     STATUS,RP0
mloop   movlw   0dh             ;send a carriage return character
        call    sendat
        movlw   3ch             ;begin main loop for data gathering and serial transmission
        call    sendat          ;these are our sync bytes to tell receiving micro a new
        movlw   3eh             ;sequence is beginning
        call    sendat
        clrf    Adcnt           ;our first time through select AN0 pin
adloop  call    adcnvrt         ;go do a conversion and send the result
        movf    Adcnt,0         ;get Adcnt into the W register
        xorlw   2               ;(# determines number of AD inputs to scan)
        btfss   STATUS,2        ;have we sampled all of the pins yet?
        goto    dopwm           ;go adjust the PWM output
        goto    mloop           ;all done go do it again
adcnvrt movf    Adcnt,0         ;get a/d count value
        movwf   Adcntw          ;put in work register
        bcf     STATUS,0        ;clear the carry flag for the upcoming rotate operations
        rlf     Adcntw,1        ;rotate left and leave the number in adcntw
        rlf     Adcntw,1        ;need to do it three times to put the count in the right
        rlf     Adcntw,1        ;position to select the next A/D pin
        movlw   0C1h            ;load the initial ADCON0 value excepting channel select
        iorwf   Adcntw,0        ;set the pin select bits we want
        movwf   ADCON0          ;set the new ADCON0 with new channels selected
        call    wait            ;wait about twenty micro seconds
        bsf     ADCON0,2        ;start conversion
        incf    Adcnt           ;increment pin counter register
adwait  btfsc   ADCON0,2        ;wait for conversion done
        goto    adwait          ;not done yet
        swapf   ADRES,0         ;conversion done, swap result nibbles into W register
        andlw   0Fh             ;mask off the upper nibble to limit number to an ascii range
        addlw   30h             ;convert to ascii character to make it visible on terminal
sendat  bsf     STATUS,RP0      ;select page one
        btfss   TXSTA,1         ;check transmit status ready to send
        goto    sendat          ;if not ready go try again
        bcf     STATUS,RP0      ;back to page 0
        movwf   TXREG           ;transmit buffer empty send new data
        return
dopwm   movf    ADRES,0         ;get the a/d conversion value
        movwf   CCPR2L          ;put the value into the PWM duty cycle register
        goto    adloop
wait    movlw   08h             ;do a wait loop of before using a/d converter
        movwf   Temp
w1      decfsz  Temp
        goto    w1
        return
        end                     ;end of program
```

FIGURE A-1: PIC16C74 DEMO SCHEMATIC



NOTES:

Note the following details of the code protection feature on PICmicro® MCUs.

- The PICmicro family meets the specifications contained in the Microchip Data Sheet.
- Microchip believes that its family of PICmicro microcontrollers is one of the most secure products of its kind on the market today, when used in the intended manner and under normal conditions.
- There are dishonest and possibly illegal methods used to breach the code protection feature. All of these methods, to our knowledge, require using the PICmicro microcontroller in a manner outside the operating specifications contained in the data sheet. The person doing so may be engaged in theft of intellectual property.
- Microchip is willing to work with the customer who is concerned about the integrity of their code.
- Neither Microchip nor any other semiconductor manufacturer can guarantee the security of their code. Code protection does not mean that we are guaranteeing the product as “unbreakable”.
- Code protection is constantly evolving. We at Microchip are committed to continuously improving the code protection features of our product.

If you have any further questions about this matter, please contact the local sales office nearest to you.

DNV Certification, Inc. USA
ANSI • RAB QMS ACCREDITED
DNV
DNV MSC The Netherlands Accredited by the RvA
ISO 9001 / QS-9000 REGISTERED FIRM

Microchip received QS-9000 quality system certification for its worldwide headquarters, design and wafer fabrication facilities in Chandler and Tempe, Arizona in July 1999. The Company’s quality system processes and procedures are QS-9000 compliant for its PICmicro® 8-bit MCUs, KEELOQ® code hopping devices, Serial EEPROMs and microperipheral products. In addition, Microchip’s quality system for the design and manufacture of development systems is ISO 9001 certified.

WORLDWIDE SALES AND SERVICE

AMERICAS

Corporate Office
2355 West Chandler Blvd.
Chandler, AZ 85224-6199
Tel: 480-792-7200 Fax: 480-792-7277
Technical Support: 480-792-7627
Web Address: http://www.microchip.com

Rocky Mountain
2355 West Chandler Blvd.
Chandler, AZ 85224-6199
Tel: 480-792-7966 Fax: 480-792-7456

Atlanta
500 Sugar Mill Road, Suite 200B
Atlanta, GA 30350
Tel: 770-640-0034 Fax: 770-640-0307

Boston
2 Lan Drive, Suite 120
Westford, MA 01886
Tel: 978-692-3848 Fax: 978-692-3821

Chicago
333 Pierce Road, Suite 180
Itasca, IL 60143
Tel: 630-285-0071 Fax: 630-285-0075

Dallas
4570 Westgrove Drive, Suite 160
Addison, TX 75001
Tel: 972-818-7423 Fax: 972-818-2924

Detroit
Tri-Atria Office Building
32255 Northwestern Highway, Suite 190
Farmington Hills, MI 48334
Tel: 248-538-2250 Fax: 248-538-2260

Kokomo
2767 S. Albright Road
Kokomo, Indiana 46902
Tel: 765-864-8360 Fax: 765-864-8387

Los Angeles
18201 Von Karman, Suite 1090
Irvine, CA 92612
Tel: 949-263-1888 Fax: 949-263-1338

New York
150 Motor Parkway, Suite 202
Hauppauge, NY 11788
Tel: 631-273-5305 Fax: 631-273-5335

San Jose
Microchip Technology Inc.
2107 North First Street, Suite 590
San Jose, CA 95131
Tel: 408-436-7950 Fax: 408-436-7955

Toronto
6285 Northam Drive, Suite 108
Mississauga, Ontario L4V 1X5, Canada
Tel: 905-673-0699 Fax: 905-673-6509

ASIA/PACIFIC

Australia
Microchip Technology Australia Pty Ltd
Suite 22, 41 Rawson Street
Epping 2121, NSW
Australia
Tel: 61-2-9868-6733 Fax: 61-2-9868-6755

China - Beijing
Microchip Technology Consulting (Shanghai) Co., Ltd., Beijing Liaison Office
Unit 915
Bei Hai Wan Tai Bldg.
No. 6 Chaoyangmen Beidajie
Beijing, 100027, No. China
Tel: 86-10-85282100 Fax: 86-10-85282104

China - Chengdu
Microchip Technology Consulting (Shanghai) Co., Ltd., Chengdu Liaison Office
Rm. 2401, 24th Floor,
Ming Xing Financial Tower
No. 88 TIDU Street
Chengdu 610016, China
Tel: 86-28-6766200 Fax: 86-28-6766599

China - Fuzhou
Microchip Technology Consulting (Shanghai) Co., Ltd., Fuzhou Liaison Office
Unit 28F, World Trade Plaza
No. 71 Wusi Road
Fuzhou 350001, China
Tel: 86-591-7503506 Fax: 86-591-7503521

China - Shanghai
Microchip Technology Consulting (Shanghai) Co., Ltd.
Room 701, Bldg. B
Far East International Plaza
No. 317 Xian Xia Road
Shanghai, 200051
Tel: 86-21-6275-5700 Fax: 86-21-6275-5060

China - Shenzhen
Microchip Technology Consulting (Shanghai) Co., Ltd., Shenzhen Liaison Office
Rm. 1315, 13/F, Shenzhen Kerry Centre,
Renminnan Lu
Shenzhen 518001, China
Tel: 86-755-2350361 Fax: 86-755-2366086

Hong Kong
Microchip Technology Hongkong Ltd.
Unit 901-6, Tower 2, Metroplaza
223 Hing Fong Road
Kwai Fong, N.T., Hong Kong
Tel: 852-2401-1200 Fax: 852-2401-3431

India
Microchip Technology Inc.
India Liaison Office
Divyasree Chambers
1 Floor, Wing A (A3/A4)
No. 11, O'Shaugnessey Road
Bangalore, 560 025, India
Tel: 91-80-2290061 Fax: 91-80-2290062

Japan
Microchip Technology Japan K.K.
Benex S-1 6F
3-18-20, Shinyokohama
Kohoku-Ku, Yokohama-shi
Kanagawa, 222-0033, Japan
Tel: 81-45-471- 6166 Fax: 81-45-471-6122

Korea
Microchip Technology Korea
168-1, Youngbo Bldg. 3 Floor
Samsung-Dong, Kangnam-Ku
Seoul, Korea 135-882
Tel: 82-2-554-7200 Fax: 82-2-558-5934

Singapore
Microchip Technology Singapore Pte Ltd.
200 Middle Road
#07-02 Prime Centre
Singapore, 188980
Tel: 65-6334-8870 Fax: 65-6334-8850

Taiwan
Microchip Technology Taiwan
11F-3, No. 207
Tung Hua North Road
Taipei, 105, Taiwan
Tel: 886-2-2717-7175 Fax: 886-2-2545-0139

EUROPE

Denmark
Microchip Technology Nordic ApS
Regus Business Centre
Lautrup hoj 1-3
Ballerup DK-2750 Denmark
Tel: 45 4420 9895 Fax: 45 4420 9910

France
Microchip Technology SARL
Parc d'Activite du Moulin de Massy
43 Rue du Saule Trapu
Batiment A - ler Etage
91300 Massy, France
Tel: 33-1-69-53-63-20 Fax: 33-1-69-30-90-79

Germany
Microchip Technology GmbH
Gustav-Heinemann Ring 125
D-81739 Munich, Germany
Tel: 49-89-627-144 0 Fax: 49-89-627-144-44

Italy
Microchip Technology SRL
Centro Direzionale Colleoni
Palazzo Taurus 1 V. Le Colleoni 1
20041 Agrate Brianza
Milan, Italy
Tel: 39-039-65791-1 Fax: 39-039-6899883

United Kingdom
Arizona Microchip Technology Ltd.
505 Eskdale Road
Winnersh Triangle
Wokingham
Berkshire, England RG41 5TU
Tel: 44 118 921 5869 Fax: 44-118 921-5820

03/01/02